JAPLA 研究会資料 2014/1/18

Jを使った連分数の計算とBessel関数
－中野先生からの年賀状を2014年の仕事始めに－

西川　利男

はじめに

中野嘉弘先生（元北海道情報大学教授）から、連分数とBessel関数のテーマを扱った、ちょっと変わった年賀状をいただいた。

連分数というのは、たとえばつぎのようなものである。

$$x = \cfrac{1}{4+\cfrac{1}{1+\cfrac{1}{27}}}$$

この値を普通の小数で表すには、逆数キー(R)のある電卓があれば、次のように簡単に求められる。

2 7 R + 1 R + 4 R => 0.2014

中野先生の問題は次のとおりである。（実物の詳細は稿末に）

ベッセル関数の一種に変形ベッセル関数In(x)（ここでnは次数）、というのがあるが、その２つの値の比が、次のようにきれいな連分数で表せる。

$$\frac{I_0(2)}{I_1(2)} = 1+\cfrac{1}{2+\cfrac{1}{3+\cfrac{1}{4+\cfrac{1}{5+\cdots}}}} = 1.43313\cdots$$

そして、先生はJを使って、従来のWatsonや林の値を越える精度でこの計算を検証され、Yahooの知恵袋に発表された、とのことであった。

このような思いがけない年賀状をいただいて、私にとっては大きなショックであった。今年は、元日から仕事始めとして、連分数とベッセル関数の問題に、私なりに取り組んでみることにした。

ここで、連分数の演算ツールとしてJが極めて強力であることが分かった。また、ベッセル関数とは、太鼓の膜の振動についての微分方程式という程度の私の認識を新たにし、勉強し直すことになった。なお、私の場合は、手元にある次の書を参考にした。

[1] **「岩波数学公式III特殊関数」森口、宇田川、一松編**(1977).
[2]　寺沢寛一「自然科学者のための数学概論(増補版)」岩波書店(1964).
[3]　木村俊一「連分数のふしぎ」講談社ブルーバックス(2012).

1．Ｊによる連分数の計算

連分数とは非常におもしろい表記法であり、木村の易しい解説書[3]には大いに触発された。

最初にあげた連分数は、次のように簡略して記されることもある。

```
 1 ; 0, 4, 1, 27
```

そして連分数の値は、Ｊでは右からの演算で次のように計算される。

```
   X =: 0 + % 4 + % 1 + % 27
   X
0.2014
```

さらに、Ｊでは引数の間に挿入の副詞 / を利用して次のように簡潔に出来る。

```
   X =: (+%) / 0, 4, 1, 27
```

これを連分数の値を求める動詞 cof と定義して

```
   cof =: (+%) /
   X =: cof 0, 4, 1, 27
```

として計算される。

一方、小数表示の値から連分数の表示に変換するには次のようにすればよい。[3]

$$0.2014=\frac{2014}{10000}=\frac{1}{\frac{10000}{2014}}=\frac{1}{4+\frac{1944}{2014}}=\frac{1}{4+\frac{1}{\frac{2014}{1944}}}=\frac{1}{4+\frac{1}{1+\frac{70}{1944}}}=\frac{1}{4+\frac{1}{1+\frac{1}{\frac{1944}{70}}}}=\frac{1}{4+\frac{1}{1+\frac{1}{27+\frac{189}{194}}}}$$

これより、動詞 n2cof としてＪのプログラムを作成した。コーディングは後述。

Ｊによる実行は次のようになる。

```
  10 n2cof 0.2014
0 4 1 27 1 3 2 1 1 1
  cof 0 4 1 27 1 1 1
0.201405
```

ところで、黄金比は、連分数表示では、次のように表される。

$$1+\frac{1}{1+\frac{1}{1+\frac{1}{1+\frac{1}{1+\cdots}}}}$$

これはＪの動詞 cof を用いて、次のように計算される。

```
  cof 1 1 1 1 1
1.6
  cof 20#1
1.61803
```

実際の値　$(1+\sqrt{5})/2$　に対して

```
   (1 + %: 5)%2
```

1.61803

　このように、連分数表示による数学は「秘められた数の素顔」を見せてくれる。[3]

２．変形ベッセル関数の比－中野の問題

　前節のような動詞 cof を使えば、先の連分数表示の変形ベッセル関数の比の値は
次のようにして求められる。

```
   >: i. 10x
 1 2 3 4 5 6 7 8 9 10
   cof >: i. 10x
 7489051r5225670
   10j8 ": cof >: i. 10x
1.43312743
```

また、このようにすれば、何桁でも求められる。

　中野先生によれば、Watson にも林にも連分数表示が載っているが、10 桁まででその後の数は不規則になっていておかしいと指摘されている。

　また、変形ベッセル関数の値は数表によって

　$I_0(2) = 2.27959$

　$I_1(2) = 1.59064$

から比を求めると、$I_0(2) / I_1(2) = 1.433127..$ となる。

３．Ｊでベッセル関数の値を求める

　ベッセル関数は、ガンマ関数を含んだ多項式展開の式として、いろいろな数学書に次のように与えられている。[1]

　第一種ベッセル関数（狭義のベッセル関数）

$$J_\nu(z) = (\frac{z}{2})^\nu \sum_{n=0}^{\infty} \frac{(-1)^n (z/2)^{2n}}{n! \Gamma(\nu+n+1)}$$

　変形されたベッセル関数

$$I_\nu(z) = (\frac{z}{2})^\nu \sum_{n=0}^{\infty} \frac{(z/2)^{2n}}{n! \Gamma(\nu+n+1)}$$

ここで、ガンマ関数はＪのプリミティブとして備えられている。したがって、Ｊでは上の式より多項べき級数の和として、直接に定義できる。つまりＪの環境ではガンマ関数の値は数表によらずとも計算により得られる。

　次にＪによる定義を示す。

NB. e.g. 0&Jn i.5 => 1 0.765198 0.223891 _0.260052 _0.39715

```
Jn =: 3 : 0"(0)
:
N =. x.
R =. i. 60
J =. ((_1)^R) % ((! R)*(! <: N + R + 1))
x =. (0.5 * y.)^(N + 2 * R)
Jx =. J * x
+/ Jx
)

In =: 3 : 0"(0)
:
N =. x.
M =. i. 100
J =. 1 % ((! M)*(! <: N + M + 1))
x =. (0.5 * y.)^(N + 2 * M)
Jx =. J * x
+/ Jx
)
```

ベッセル関数 J_0の数表を計算してみよう。

```
   (3j1 12j6) ":(],. 0&Jn) X    NB. J0
0.0     1.000000
0.5     0.938470
1.0     0.765198
1.5     0.511828
2.0     0.223891
2.5   _0.048384
3.0   _0.260052
3.5   _0.380128
4.0   _0.397150
4.5   _0.320543
5.0   _0.177597
5.5   _0.006844
```

変形されたベッセル関数 I_0の数表は次のようになる。

```
   (3j1 12j6) ":(],. 0&In) X
```

0.0	1.000000
0.5	1.063483
1.0	1.266066
1.5	1.646723
2.0	2.279585
2.5	3.289839
3.0	4.880793
3.5	7.378203
4.0	11.301922
4.5	17.481172
5.0	27.239872
5.5	42.694645

plot ルーチンにより、ベッセル関数を図示してみよう。（左図は J0 右図は J1）

```
 load 'plot'
 X =: (i.1001) % 100
 plot X; 0&Jn X
```



変形ベッセル関数の図示は次のようになる。



Bessel関数の I0(2) と I1(2) の値の比 r は、Cambridge Uni.
G.N.Watson の大著 Bessel関数(1922)の末尾の付表 p.699ではrW =
1.432969488であり、日本の 九大教授・林桂一の有名な高等関数表
(岩波書店,1941) p.157では、rH = 1.433127546であるが、実は奇妙
な特徴がある。数値の連分数展開が 1 ; 2 3 4 5 となるのだ。
普通の書式で書けば、
r5 = 225 / 157 = 1 + 1/(2+ 1/(3+ 1/(4+ 1/(5)))) かな！
詳しくは、Watson値では 4 まで、林の値では 5までが成立する。
問題は、この奇妙な連分数列が、どこまで続くであろうか ? である。

私のテストでは、比の値 rN = 1393 / 972 (= 1.433127572) なら、
連分数列 = 1 ; 2 3 4 5 6 であり、rN= 9976 / 6961(1.433127424)
ならば、1 ; 2 3 4 5 6 7 。rN = 81201/56660 (1.433127427)では
1 ; 2 3 4 5 6 7 8 etc. である。今までの最終結果では、rN =
764582487395121 / 533506283627401 (1.433127427.......)で、
1 ; 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 が得られた。
まこと、見事なもので、小数点下9桁の先の誤差のゴミ同然の処から、
こんな綺麗な結果が顔を出すとは! もって、お正月の贈り物とする。

2014(平成26年) 午年元旦、 中野嘉弘 (91翁)

```
NB. Bessel Function and Continuous Fraction

wr =: 1!:2&2

NB.   Nakano 2014/1/2

NB.   r2 =: 1 + 1 %((2)) + 1 % ((3)) + 1 % ((4)) + 1 % ((5))
NB.   r2
NB. 1.43312
NB.   r2 =: 1 + 1 %((2)) + 1 % ((3)) + 1 % ((4)) + 1 % ((5)) + 1 % ((6))
NB.   r2
NB. 1.43313
NB.   r2 =: 1 + 1 %((2)) + 1 % ((3)) + 1 % ((4)) + 1 % ((5)) + 1 % ((6)) + 1 %
((7))
NB.   r2
NB. 1.43313
NB.   r2 =: 1 + 1 %((2)) + 1 % ((3)) + 1 % ((4)) + 1 % ((5)) + 1 % ((6)) + 1 %
((7x))
NB.   r2
NB. 9976r6961
NB.   12j10 ": r2
NB. 1.4331274242

NB. Continuos Fraction to Number - J Introduction and Dictionary p.46
   a =: 2 3 5 7 11
   cof =: (+%) /
NB.   a
NB. 2 3 5 7 11
NB.   11j9 ": cof a
NB. 2.313036690

NB.   >: i. 10x
NB. 1 2 3 4 5 6 7 8 9 10
NB.   cof >: i. 10x
NB. 7489051r5225670
NB.   10j8 ": cof >: i. 10x
NB. 1.43312743
```

```
NB.   cof >: i. 17x
NB. 764582487395121r533506283627401
NB.   22j20 ": cof >: i. 17x
NB. 1.43312742672231175832

NB.   24j22 ": cof >: i. 20x
NB. 1.4331274267223117583172

NB. Number to Continuous Fraction ==================
NB. 木村俊一「連分数のふしぎ」講談社ブルーバックス p.43
NB. 6 n2cof 1.3114754 => 1 3 4 1
NB. 6 n2cof 1.2345678 => 1 4 3 1
n2cof =: 3 : 0
:
N =. x.
X =. y.
XS =. <. X              NB. 整数部分の最初の値
CF =. ''
I =. 0
while. I < N
  do.
NB.     wr '------------'
NB.     wr 'I=', ": I
     CFX =. <. X        NB. 整数部分
     XF =. X - CFX      NB. 小数部分
     RXF =. % XF        NB. 小数部分の逆数
     CF =. CF, CFX      NB. 整数部分の次の桁への更新
     X =. RXF           NB. 小数部分を次の繰り返しの値とする
     if. (I > 0) *. (CFX = XS) do. break. end.
     I =. I + 1
  end.
CF
)
```

```
NB. Bessel Function by T.N 2014/1/3 ===============

NB. e.g.
NB.  J0 i. 5 => 1 0.765198 0.223891 _0.260052 _0.39715

J0 =: 3 : 0"(0)
R =. i. 60
J =. ((_1)^R) % (*: ! R)
x =. (0.5 * y.)^(2*R)
Jx =. J * x
+/ Jx
)

J1 =: 3 : 0"(0)
R =. i. 60
J =. ((_1)^R) % ((! R)*(! >:R))
x =. (0.5 * y.)^(1 + 2*R)
Jx =. J * x
+/ Jx
)

NB. e.g. 0&Jn i.5 => 1 0.765198 0.223891 _0.260052 _0.39715
Jn =: 3 : 0"(0)
:
N =. x.
R =. i. 60
J =. ((_1)^R) % ((! R)*(! <: N + R + 1))
x =. (0.5 * y.)^(N + 2 * R)
Jx =. J * x
+/ Jx
)

In =: 3 : 0"(0)
:
N =. x.
M =. i. 100
J =. 1 % ((! M)*(! <: N + M + 1))
```

```
x =. (0.5 * y.)^(N + 2 * M)
Jx =. J * x
+/ Jx
)
```